

## 0.S3 ジュニア版 特別章(3) 格子とツリーのプログラミング方法

ここでは、多重ループの応用として、2項や3項の格子やツリーをプログラムします。スプレッドシート式の思考方法やプログラミング法を脱して、まずは表の大枠を確保するのに0を埋めこんで、そのすべてを欠損値に変換することから始めましょう。

### 0 から始めて表の確保する

例えば、縦9横5（すなわち9行5列）の零行列を作成しておいて、それを欠損値のドット表示に変換することによって、これから数値を埋め込んでいく表の大枠を作ります。

#### プログラム

```
new;  
cls;  
m=miss(zeros(5,5),0);  
print m;
```

#### 画面表示

```
      .      .      .      .      .  
      .      .      .      .      .  
      .      .      .      .      .  
      .      .      .      .      .  
      .      .      .      .      .
```

#### プログラム各行の説明

- 1 行目 メモリを初期化する。
- 2 行目 スクリーンをクリアする。
- 3 行目 5行5列の零行列を作成しておき、その行列の要素が0のものを組込み関数 miss によって欠損値のドットに変換する。その結果を変数mに代入する。
- 4 行目 変数mを画面表示する。

なお、3行目はm=zeros(5,5);として変数mに5行5列の零行列を代入しておいて、それを組込み関数missによってm=miss(m,0);とすることによって、変数mの要素のうち0であるものを欠損値のドットに変換した結果をあらためて変数mに代入するのと同じことです。GAUSSには、スプレッドシートのように枠がありません。その代わりに、例えばこのように表がイメージできるようにドットで表しておきましょう。

### 行列の行と列のインデックス

プログラムの世界では通常、行はi列はjでその行列のインデックスを表します。私たちもその考え方を極力変えることなく用いて、これからのプログラムを考えることにしましょう。まずは練習として、表の行の1行目に横に1,2,3,4,5と代入してみましょう。つまり

今、上と同じ 5 行 5 列の表の大枠を考えているものだとすると、1 行目の 1 行に 1 列目から順に 1,2,3,4,5 と 5 列目まで埋めることになります。

#### プログラム

```
new;  
cls;  
m=miss(zeros(5,5),0);  
j=1;  
do while j<=5;  
    m[1,j]=j;  
    j=j+1;  
endo;  
print m;
```

#### 画面表示

1.0000000	2.0000000	3.0000000	4.0000000	5.0000000
.	.	.	.	.
.	.	.	.	.
.	.	.	.	.
.	.	.	.	.

#### プログラム各行の説明

- 1 行目 メモリを初期化する。
- 2 行目 スクリーンをクリアする。
- 3 行目 5 行 5 列の零行列を作成しておき、その行列の要素が 0 のものを組込み関数 miss によって欠損値のドットに変換する。その結果を変数 m に代入する。
- 4 行目 列のインデックス番号 j を 1 に設定しておく。
- 5 行目 j が 5 (すなわち変数 m の列数) 以下の間はこの Do ループを行なう。
- 6 行目 変数 m の 1 行目の各 j 列目に j を代入していく。
- 7 行目 j を 1 増やしてあらためて j とする。
- 8 行目 この Do ループの終わりの部分。ここまでを繰り返す。
- 9 行目 変数 m を画面表示する。

すなわち、j が 1 の 1 列目から m の列数まで、変数 m の 1 行目に 1 列目には 1 を 2 列目には 2 を、これを 5 列目まで繰り返して、この Do ループを脱して、最終的に変数 m を画面表示させるプログラムです。行を表すインデックスはこの場合 1 で固定されていて、列を表す j を 1 から m の列数である 5 まで 1 つずつ動かしています。ちょうど

	1 列目	2 列目	3 列目	4 列目	5 列目
1 行目	m[1,1]=1,	m[1,2]=2,	m[1,3]=3,	m[1,4]=4,	m[1,5]=5

となる 1 から 5 までの部分を j と置いたものです。ループで j=1 から 5 までインデックス番号を動かすことによって、数字を代入すべき場所を動かしているのと同時に、代入すべき数値も（この場合偶然に列番号と一致しているが）j=1 から 5 まで動かしているのです。

今度は、いままで 1 行目に固定していた行もインデックス番号を用いて動かしてみる例を考えます。1 列目は 1 行目まで、2 列目は 2 行目まで、これを繰り返し 5 列目は 5 行目まで列番号と同じ数を表に代入することをやってみます。

#### プログラム

```
new;
cls;
m=miss(zeros(5,5),0);
j=1;
do while j<=5;
    i=1;
    do while i<=j;
        m[i,j]=j;
        i=i+1;
    endo;
    j=j+1;
endo;
print m;
```

#### 画面表示

1.0000000	2.0000000	3.0000000	4.0000000	5.0000000
.	2.0000000	3.0000000	4.0000000	5.0000000
.	.	3.0000000	4.0000000	5.0000000
.	.	.	4.0000000	5.0000000
.	.	.	.	5.0000000

#### プログラム各行の説明

- 1 行目 メモリを初期化する。
- 2 行目 スクリーンをクリアする。
- 3 行目 5 行 5 列の零行列を作成しておき、その行列の要素が 0 のものを組込み関数 miss

によって欠損値のドットに変換する。その結果を変数mに代入する。

- 4 行目 列のインデックス番号  $j$  を 1 に設定しておく。
- 5 行目  $j$  が 5 (すなわち変数  $m$  の列数) 以下の間はこの外側の Do ループを行なう。
- 6 行目 行のインデックス番号  $i$  を 1 に設定しておく。
- 7 行目  $i$  が  $j$  以下の間はこの内側の Do ループを行なう。
- 8 行目 変数  $m$  の  $i$  行  $j$  列目に列番号と同じ  $j$  を代入していく。
- 9 行目  $i$  を 1 増やしてあらためて  $i$  とする。
- 10 行目 この内側の  $i$  に関する Do ループの終わりを表す部分。
- 11 行目  $j$  を 1 増やしてあらためて  $j$  とする。
- 12 行目 この外側の  $j$  に関する Do ループの終わりを表す部分。
- 13 行目 変数  $m$  を画面表示する。

少し複雑ですが、内側の  $i$  に関する Do ループが挿入されていて、行を表すインデックス番号の  $i$  もそれぞれの列  $j$  に対して 1 ずつまされいていると考えてください。ただし、この  $i$  は最大  $j$  までしか行かないという意味で  $\text{do while } i \leq j$ ; というふうに最大行数 5 ではなくて  $j$  がきています。第  $j$  列では、最大  $j$  行目までしか数字は入らないということです。

## 2 項格子 (簡単なケース)

ここで、上の  $i$  と  $j$  を動かす 2 重ループ (各  $j$  列に対して  $i$  を動かすループ) と同じものを用いて、今度は  $j$  を代入するのではなくて 2 項格子にしたがうダイナミックな数を代入してみましょう。今、初期値  $S = 100$  に対して、水平に左に移動すれば  $u = 1.1$  をかけたもの、そうではなくて下降して左に移動する場合  $d = 1/u$  をかけたものがくるような、簡単な 2 項格子モデルを考えよう。すなわち、求めたいのは

$S$	$Su$	$Su^2$	$Su^3$	$Su^4$
	$Sd$	$Sud$	$Su^2d$	$Su^3d$
		$Sd^2$	$Sud^2$	$Su^2d^2$
			$Sd^3$	$Sud^3$
				$Sd^4$

である。これをプログラム化するため、 $u$  と  $d$  の両方が各項にあらわれるように書き直すと

$Su^0d^0$	$Su^1d^0$	$Su^2d^0$	$Su^3d^0$	$Su^4d^0$
	$Su^0d^1$	$Su^1d^1$	$Su^2d^1$	$Su^3d^1$
		$Su^0d^2$	$Su^1d^2$	$Su^2d^2$
			$Su^0d^3$	$Su^1d^3$
				$Su^0d^4$

となる。ここでは何も  $u$  または  $d$  の項がない部分は  $u^0 = 1$  と  $d^0 = 1$  であることと、 $u$  と  $d$  が単独でくるところは 1 乗と考えてすべてを書きなおしている。乗数の和が 1 列目は 0、2

列目は1、3列目は2、4列目は3、5列目は4にすべての項がなっていることに着目しよう。パターンがわかれば、あとのプログラム化は簡単である。すなわち、第j列目は常にuまたはdにかかる乗数がj-1の和になっているのである。なお、uの乗数はj-1から1ずつその列を1つ下にいくごとに減って最終的に0になっている。他方、dの乗数は1ずつその列を1つ下にいくごとに増えて最終的にj-1になっている。

#### プログラム

```
new;
cls;
S=100; u=1.1; d=1/u;
m=miss(zeros(5,5),0);
j=1;
do while j<=5;
    i=1;
    do while i<=j;
        m[i,j]=S*u^(j-i)*d^(i-1);
        i=i+1;
    endo;
    j=j+1;
endo;
print m;
```

#### 画面表示

100.00000	110.00000	121.00000	133.10000	146.41000
.	90.909091	100.00000	110.00000	121.00000
.	.	82.644628	90.909091	100.00000
.	.	.	75.131480	82.644628
.	.	.	.	68.301346

#### プログラム各行の説明

- 1行目 メモリを初期化する。
- 2行目 スクリーンをクリアする。
- 3行目 初期値Sは100、uは1.1そしてdはuの逆数1/uに設定しておく。
- 4行目 5行5列の零行列を作成しておき、その行列の要素が0のものを組込み関数 miss によって欠損値のドットに変換する。その結果を変数mに代入する。
- 5行目 列のインデックス番号jを1に設定しておく。
- 6行目 jが5（すなわち変数mの列数）以下の間はこの外側のDoループを行なう。
- 7行目 行のインデックス番号iを1に設定しておく。
- 8行目 iがj以下の間はこの内側のDoループを行なう。

- 9 行目 変数  $m$  の  $i$  行  $j$  列目に  $S u^{j-i} d^{i-1}$  を代入していく。
- 10 行目  $i$  を 1 増やしてあらためて  $i$  とする。
- 11 行目 この内側の  $i$  に関する Do ループの終わりを表す部分。
- 12 行目  $j$  を 1 増やしてあらためて  $j$  とする。
- 13 行目 この外側の  $j$  に関する Do ループの終わりを表す部分。
- 14 行目 変数  $m$  を画面表示する。

これまで、変数  $m$  の  $i$  行  $j$  列目には  $j$  を代入していたが、ここでは 2 項格子の意味のある数値を計算して、パターン化して代入している。つまり、第  $j$  列について一番上の  $u$  の乗数は 1 を引いた  $j - 1$  である。その下は  $j - 2$ 、これを繰り返す。これを  $i$  行目の値を見る観点から考えると、 $j - i$  を  $i = 1$  から順に 1 つずつ  $i$  を増していることにあたる。他方、 $d$  の乗数は一番上から順に 0 から 1 ずつ増えていく。一番上は 1 行目であるから、今内側のループで  $i$  を動かしているので、 $i - 1$  から順に行数が 1 増えるごとにこの数も増えていく。実際、 $u$  の乗数  $j - i$  と  $d$  の乗数  $i - 1$  とを合計した数は  $j - 1$  になっている。これはパターンを見たときの各列  $j$  の乗数の和に等しくなっている。これで、第  $j$  列に対して、 $i$  を 1 から  $j$  まで動かすときの計算パターンが決まったわけである。これを前の  $j$  を代入するところにもってきて同じ計算をさせれば、求める 2 項格子の計算の完成である。これを procedure 化して、 $S=100$  のところを 0 期、その隣の 2 列目を 1 期、などとして、0 期から  $n=4$  期までの 2 項格子を求める procedure を作成してみよう。

#### プログラム

```
new;
cls;
S=100; u=1.1; d=1/u;
n=4;
call lattice2(S,u,d,n);

proc lattice2(S,u,d,n);
    local m,i,j;
    m=miss(zeros(n+1,n+1),0);
    j=1;
    do while j<=cols(m);
        i=1;
        do while i<=j;
            m[i,j]=S*u^(j-i)*d^(i-1);
            i=i+1;
        endo;
        j=j+1;
    endo;
endproc;
```

```

        j=j+1;
    endo;
    print/rz seqa(0,1,n+1)';
    print m;
    retp(m);
endp;

```

画面表示

0	1	2	3	4
100.00000	110.00000	121.00000	133.10000	146.41000
.	90.909091	100.00000	110.00000	121.00000
.	.	82.644628	90.909091	100.00000
.	.	.	75.131480	82.644628
.	.	.	.	68.301346

## 2 項格子（通常のケース）

これまでは 2 項格子の簡単なケースであったが、左に平行に行くところを上昇  $u$ 、下降するところを下降  $d$  として同じことを計算してみよう。結果は同じになるが、今度はかなりインデックスの扱いが複雑になる。ここでは、これまでと同じように第  $j$  列目に対して、内側の Do ループのインデックス  $i$  を 1 から 1 つずつ増やす方法をとることにしよう。

プログラム

```

new;
cls;
S=100; u=1.1; d=1/u;
n=4;
call lat2(S,u,d,n);

proc lat2(S,u,d,n);
    local m,i,j;
    m=miss(zeros(2*n+1,n+1),0);
    j=1;
    do while j<=cols(m);
        i=1;
        do while i<=2*j-1;
            m[n+1-j+i,j]=S*u^(j-1-(i-1)/2)*d^((i-1)/2);
            i=i+2;
        end;
        j=j+1;
    end;
end;

```

```

        endo;
        j=j+1;
    endo;
    print/rz seqa(0,1,n+1)';
    print m;
    retp(m);
endp;

```

画面表示

	0	1	2	3	4
	.	.	.	.	146.41000
	.	.	.	133.10000	.
	.	.	121.00000	.	121.00000
	.	110.00000	.	110.00000	.
100.00000	.	.	100.00000	.	100.00000
	.	90.909091	.	90.909091	.
	.	.	82.644628	.	82.644628
	.	.	.	75.131480	.
	.	.	.	.	68.301346

### 3 項格子

今度は、上昇は  $u$  倍、水平は 1 倍、そして下降は  $d$  倍となる 3 項格子を考えよう。ただし、 $d$  は  $u$  の逆数に必ず一致するものとする。そうすると、計算結果は

				$Su^4$
			$Su^3$	$Su^3$
		$Su^2$	$Su^2$	$Su^2$
	$Su$	$Su$	$Su$	$Su$
$S$	$S$	$S$	$S$	$S$
	$Sd$	$Sd$	$Sd$	$Sd$
		$Sd^2$	$Sd^2$	$Sd^2$
			$Sd^3$	$Sd^3$
				$Sd^4$

というふうな並びになるはずである。なぜならば、 $d=1/u$  なので  $du=1$  になるからだ。プログラムでは、列ではなくて行  $i$  に着目して、先に真ん中の同じ  $S$  の行を先に埋めておいてから、上半分の  $u$  からなる三角形の部分と下半分の  $d$  からなる三角形の部分に分割して、それぞれについて、1 乗の  $u$ 、2 乗の  $u$ 、これを繰り返して各行ごとに代入する。同様にし



て、下の三角形についても 1 乗の  $d$ 、2 乗の  $d$ 、これを繰り返す。

#### プログラム

```
new;
cls;
S=100; u=1.1; d=1/u;
n=4;
call lat3(S,u,d,n);

proc lat3(S,u,d,n);
    local m,i,j;
    m=miss(zeros(2*n+1,n+1),0);
    j=1;
    do while j<=n+1;
        m[n+1,j]=S*1;
        j=j+1;
    endo;
    i=n;
    do while i>=1;
        j=n+2-i;
        do while j<=n+1;
            m[i,j]=S*u^(n+1-i);
            j=j+1;
        endo;
        i=i-1;
    endo;
    i=n+2;
    do while i<=2*n+1;
        j=i-n;
        do while j<=n+1;
            m[i,j]=S*d^(i-n-1);
            j=j+1;
        endo;
        i=i+1;
    endo;
    print/rz seqa(0,1,n+1)';
    print m;
```

```
ret p(m);
end p;
```

画面表示

	0	1	2	3	4
.	.	.	.	.	146.41000
.	.	.	.	133.10000	133.10000
.	.	.	121.00000	121.00000	121.00000
.	110.00000	110.00000	110.00000	110.00000	110.00000
100.00000	100.00000	100.00000	100.00000	100.00000	100.00000
.	90.909091	90.909091	90.909091	90.909091	90.909091
.	.	82.644628	82.644628	82.644628	82.644628
.	.	.	75.131480	75.131480	75.131480
.	.	.	.	.	68.301346

## 2 項ツリー (YES と NO の判断分岐)

ツリーは、一般に、格子よりもプログラムは複雑になる。まずは、1 と 0 からなるツリーを作成し、1 なら YES を 0 なら NO を割り当てて、その判断分岐のグラフを作ってみよう。ここでは、あとの計算も考えて、先に 1 と 0 を割り当てておいて、1 なら YES という文字列を 0 なら NO という文字列になるように変換した結果を表示させることにする。

プログラム

```
new;
cls;
n=2;
call treeYESNO(n);

proc treeYESNO(n);
    local label1,label0,m,onezero,i,j;
    label1={"YES"};
    label0={" NO"};
    m=miss(zeros((2^n)*2-1,n+1),0);
    onezero=1 | 0;
    j=1;
    m[2^(n+1-j),j]=0;
    j=2;
    do while j<=n+1;
```

```

i=1;
do while i<=2^(j-1);
    onezero=onezero | 1 | 0;
    m[2^(n+1-j)+2^(n+2-j)*(i-1),j]=onezero[i];
    i=i+1;
end;
j=j+1;
end;
j=1;
do while j<=cols(m);
    i=1;
    do while i<=rows(m);
        if m[i,j]==0;
            m[i,j]=label0;
        elseif m[i,j]==1;
            m[i,j]=label1;
        endif;
        i=i+1;
    end;
    j=j+1;
end;
print/rz seqa(0,1,n+1)';
print $m;
retp(m);
endp;

```

画面表示

0	1	2
.	.	YES
.	YES	.
.	.	NO
NO	.	.
.	.	YES
.	NO	.
.	.	NO

## 2 項ツリーモデル

今度は、上昇 YES か下降 NO の分岐から得られる Payoff を考えるモデルを考える。なお、YES と NO の出方は上のものと同じとする。YES なら  $p$  の割合で減るものとする。NO であれば変化はないものとする。これは、かなり複雑なパターン計算を必要とする。

### プログラム

```
new;
cls;
S=100; p=0.1; n=2;
call btree(S,p,n);

proc btree(S,p,n);
    local label0,label1,m,m2,m3,onezero,i,j,index;
/* Yes(1) or No(0) */
    label1={"YES"};
    label0={" NO"};
    m=miss(zeros((2^n)*2-1,n+1),0);
    onezero=1 | 0;
    j=1;
    m[2^(n+1-j),j]=0;
    j=2;
    do while j<=n+1;
        i=1;
        do while i<=2^(j-1);
            onezero=onezero | 1 | 0;
            m[2^(n+1-j)+2^(n+2-j)*(i-1),j]=onezero[i];
            i=i+1;
        endo;
        j=j+1;
    endo;
    j=1;
    do while j<=cols(m);
        i=1;
        do while i<=rows(m);
            if m[i,j]==0;
                m[i,j]=label0;
            elseif m[i,j]==1;

```

```

        m[i,j]=label1;
    endif;
    i=i+1;
end;
j=j+1;
end;
print/rz seqa(0,1,n+1)';
print $m;
/* Amount left */
m2=miss(zeros(2^n,n+1),0);
m2[:,1]=zeros(2^n,1);
j=n+1;
do while j>=2;
    index=ones(2^(n+1-j),1) | zeros(2^(n+1-j),1);
    i=1;
    do while i<=(2^n)/(2^(n+1-j)*2)-1;
        index=index | ones(2^(n+1-j),1) | zeros(2^(n+1-j),1);
        i=i+1;
    end;
    m2[:,j]=index;
    j=j-1;
end;
m2=S*(1-p)^cumsumc(m2)';
m3=miss(zeros((2^n)*2-1,n+1),0);
j=1;
do while j<=n+1;
    i=1;
    do while i<=2^(j-1);
        m3[2^(n+1-j)+2^(n+2-j)*(i-1),j]=m2[1+2^(n+1-j)*(i-1),j];
        i=i+1;
    end;
    j=j+1;
end;
print;
print "Amount left:";
print/rz seqa(0,1,n+1)';

```

```

print/rz m3;
retp(m3);
endp;

```

画面表示

0	1	2
.	.	YES
.	YES	.
.	.	NO
NO	.	.
.	.	YES
.	NO	.
.	.	NO

Amount left:

0	1	2
.	.	81
.	90	.
.	.	90
100	.	.
.	.	90
.	100	.
.	.	100

ここで procedure の前半部分はその前の YES と NO のものと同じプログラムである。後半部分は、 $m2=S*(1-p)^{\text{cumsumc}(m2)}$  の行の前に仮に print m2; として m2 の内容を表示してやると何をやっているのかわかるだろう。すなわち、

0.00000000	1.00000000	1.00000000
0.00000000	1.00000000	0.00000000
0.00000000	0.00000000	1.00000000
0.00000000	0.00000000	0.00000000

というような 1 と 0 のからなるインデックス行列を作成していることになる。1 は YES であり 0 は NO である。各行がその YES と NO の履歴になる。1 行目は 0,1,1 だから NO から始めて YES, YES ということであり、2 行目は NO から始めて YES, NO ということである。このように各行を横に見てやるとその YES と NO の出具合がわかる。なお、これを縦

方向、各列で見てやると、1列目はすべて0、2列目は上半分が1下半分が0となっている。次の列は、前列の上半分のそのまた半分が1で、あと残りの半分が0である。さらに前列の下半分のそのまた半分が1で、その残り半分が0である。これを繰り返す。最終列では YES と NO (すなわち 1 と 0) が交互にくることになる。これをプログラムしたものが、procedure 後半の今 print m2;を仮に挿入してみた部分よりも前のところである。

さらに、このインデックス行列の「行ごとの」累積和を計算すると

0.00000000	1.00000000	2.00000000
0.00000000	1.00000000	1.00000000
0.00000000	0.00000000	1.00000000
0.00000000	0.00000000	0.00000000

となる。これは、GAUSS では cumsumc(m2')によって計算されるものである。つまり、インデックス行列を転置して列ごとの累積和を計算してやり、もとのインデックス行列と同じ形に直すために、その結果自体をさらに転置させているのである。行の横方向に見てやって、0は1が一回もきていないこと、1は1が1回、2は1が2回きていることを表している。最終的に、 $S \cdot (1-p)^{\text{cumsumc}(m2')}$ としてやって、1からpだけの割合を除いた残りの部分をこの累積和で累乗したものに初期値Sをかけると、残りの量の行列ができる。これを表示してもよいのだが、最終行を除き同じものが重なって含まれていて、なおかつツリーの形をなしていない。そこで、それを procedure の最後の部分で行列 m3 の中にピックアップし、 $(2^n) \cdot 2-1$  行  $n+1$  列に重なりがないように代入していくのである。

```

j=1;
do while j<=n+1;
    i=1;
    do while i<=2^(j-1);
        m3[2^(n+1-j)+2^(n+2-j)*(i-1),j]=m2[1+2^(n+1-j)*(i-1),j];
        i=i+1;
    endo;
    j=j+1;
endo;

```

上のように、外側の j ループで各 j 列について見ていく。内側の i ループは上で計算した重なりを含む結果が1行目から始まるため、 $i = 1$  から始まっている。第 j 列のノードの数は、最初が1つ、次が2つ、その次が4つと続くから、 $2^{j-1}$  まで数えればよい。重なりを含んだ計算結果 m2 からピックアップして、ツリーの形をなした変数行列 m3 に代入し直すのだが、問題は行のところになる。なお、m2 の最初はいつも1になる。実際、 $(i-1)$  のところが1から回るので、常にその項は0から始まることになり、 $1 +$  のところが残り、これでよい。例えば、2列目は1行目と3行目の計2つをピックアップする必要がある、3列目は、この場合( $n=2$ ) j が3に対して  $2^{(n+1-j)}$  の累乗のところは0になるので、この部

分は2の0乗で1となる。全体としては1,2,3,4と順番にm2の行をピックアップしていくことになっている。この部分はすべての自然数nについて成り立つように書いているのでこのように複雑になっている。これをj列ごとに $m3[2^{(n+1-j)}+2^{(n+2-j)}*(i-1),j]$ に代入していく。n = 2に対して、1列目は4行目だけ、2列目は2行目と6行目、3列目は1,3,5,7行目だから上の式は一致している。複雑な累乗数になっているが、出発点+増分の計算をしているだけである。このように最終的にツリーの形に変形することで計算は完成する。途中複雑な計算をしているように見えるが、足し算掛け算そして累乗の数を使って、j列目に対してnが決まっているときの配列の場所を式で表しているわけである。ツリーのノードとノードの間隔があいているものを作りたい場合や、上から展開していくツリーを作りたい場合でも、どのようにしてパターンがあるかぎり足し算掛け算そして累乗の数でインデックスは作成できるものだと考えて、式を考えてあてはめて確認する作業を行なっているのです。

このように、格子やツリーは2次元のiとjの配列のインデックスを用いてプログラムしていきます。上のやり方は、ほんの1つのやり方にすぎません。パターンがあるかぎり、特にそのパターンが上下対称の場合、必ずiとjの配列表現が可能です。2項格子が下から上へ展開していようが、2項ツリーが上から下へと展開していようが、同様にパターンを足し算や掛け算それに累乗の計算でiとjの配列で表現してプログラムできます。

発展

Luenberger	2項格子	5.3b	11.1a
	3項格子	13.7d	16.3a
	2重3項格子	16.5a	
	2項ツリー	5.3c	16.6a